@强制输出
@输出名("木鱼")
类 木鱼 : 窗口

	@布局配置([[根布局=真,宽度=-1,高度=-1]])
	变量 相对布局1 : 相对布局
	@布局配置([[父布局=相对布局1,宽度=-1,高度=-1,拉伸方式=0]])
	变量 我的背景 : 图片框
	@布局配置([[父布局=相对布局1,宽度=-1,高度=-1,@位于某组件之下=2]])
	变量 相对布局4 : 相对布局
	@布局配置([[父布局=相对布局4,宽度DP=180,高度DP=180,@位于布局中间=真,ID=1]])
	变量 相对布局3 : 相对布局
	@布局配置([[父布局=相对布局3,宽度DP=150,高度DP=150,@位于布局中间=真,拉伸方式=0]])
	变量 我的木鱼 : 图片框
	@布局配置([[父布局=相对布局4,内容="功德 +1",粗体=真,字体大小=22,@与某组件右边平齐=1,@与某组件顶部平齐=1,可视=假,透明度=0.76]])
	变量 我的功德 : 文本框
	@布局配置([[父布局=相对布局1,宽度=-1,高度DP=45,左内边距DP=20,右内边距DP=20]])
	变量 相对布局2 : 相对布局
	@布局配置([[父布局=相对布局2,单行显示=假,字体大小=16,字体颜色=-1,粗体=真,对齐方式=17,内容="我的功德：0",内边距DP=5,@位于布局中间=真,透明度=0.79]])
	变量 我的功德数 : 文本框


	变量 当前背景:整数
	变量 当前木鱼:整数
	变量 当前音效:整数
	变量 自定义颜色:整数=黑色
	变量 功德数:整数
	变量 我的功德总数:整数
	变量 木鱼动画 : 组件动画=偏移动画.新建(1,1,1,-10)
	变量 功德动画 : 组件动画=偏移动画.新建(1,1,1,-200)
	事件 木鱼:创建完毕()
		订阅事件()
		本对象.屏幕方向 = 1
		本对象.显示标题栏 =假
		木鱼动画.播放时间 = 100
		功德动画.播放时间 = 300
		当前背景=全局配置.个人数据.取整数("木鱼背景")
		当前木鱼=全局配置.个人数据.取整数("木鱼皮肤")
		当前音效=全局配置.个人数据.取整数("木鱼音效")
		我的功德总数=全局配置.个人数据.取整数("木鱼功德数")
		我的功德数.内容 = "我的功德："+我的功德总数.到文本()
		置圆角边框(我的功德数,2,白色,0,15,0,15)
		判断背景()
	结束 事件
	方法 判断背景()
		假如 当前背景
			是 1
				本对象.状态栏字体黑色 = 真
				本对象.状态栏颜色 = 文本到颜色值("#4E0A06")
				本对象.导航栏颜色 = 文本到颜色值("#410805")
			否则
				本对象.状态栏字体黑色 = 假
				本对象.状态栏颜色 = 自定义颜色
				本对象.导航栏颜色 = 自定义颜色
				相对布局1.背景颜色 = 自定义颜色
				我的功德.字体颜色 = 白色
		结束 假如
		我的背景.位图对象 = 附加资源.取位图对象(本对象,"Divination/功德/木鱼/背景/"+当前背景.到文本())
		我的木鱼.位图对象 = 附加资源.取位图对象(本对象,"Divination/功德/木鱼/皮肤/"+当前木鱼.到文本())
	结束 方法
	事件 相对布局4:被触摸(来源事件 : 触摸事件) : 逻辑型
		如果 来源事件.动作==触摸动作.按下 则
			敲击()
		结束 如果
		返回 真
	结束 事件
	方法 敲击()
		媒体操作.播放assets音频(本对象,"Divination/功德/木鱼/音效/"+当前音效.到文本())
		自增(功德数,1)
		自增(我的功德总数,1)
		全局配置.个人数据.置入("木鱼功德数",我的功德总数)
		全局配置.个人数据.写出到文件(取数据目录路径()+"/配置文件/个人数据/个人数据.json")
		我的功德数.内容 = "我的功德："+我的功德总数.到文本()
		我的木鱼.播放动画(木鱼动画)
		我的功德.可视 = 真
		我的功德.播放动画(功德动画)
	结束 方法
	事件 我的功德:动画播放完毕()
		我的功德.可视 = 假
	结束 事件
	事件 我的功德数:被单击()
		自增(当前木鱼,1)
		判断背景()
	结束 事件

结束 类